



2019



Total Sales

\$3.55M

-28.77% ▼

Total Profit

\$1.98M

-29.09% ▼

Unique Customer

1697

-27.17% ▼

Total Customer

1795

-24.04% ▼

Year


○ 2017

○ 2018


● 2019

○ 2020


LastYear Sales, Sales and Profit by Month




Sales By Store Country




Online store customer order count by country



Customer buy more than once by country



Total stores



Delivery days by online store

Country	1	2	3	5	7	10
Australia	18.18%	43.18%	67.05%	97.73%	100.00%	
Canada	20.13%	45.64%	73.15%	94.63%	99.33%	100.00%
France	20.34%	45.76%	74.58%	93.22%	98.31%	100.00%
Germany	14.62%	41.54%	67.69%	93.85%	100.00%	
Italy	23.53%	41.18%	64.71%	94.12%	100.00%	
Netherlands	18.97%	46.55%	65.52%	94.83%	100.00%	
United Kingdom	11.54%	48.72%	69.23%	94.87%	100.00%	
United States	13.35%	41.68%	69.10%	97.07%	100.00%	
Total	15.45%	43.25%	69.36%	95.91%	99.82%	100.00%


Most sold product

Category	Rank	Sales Amount	Profit Amount	%profit
Computers	1	\$1,290,782.31	\$718,810.048	55.69%
Home Appliances	2	\$645,841.52	\$357,861.653	55.41%
Cell phones	3	\$631,763.92	\$342,298.8803	54.18%
TV and Video	4	\$403,851.45	\$228,633.3045	56.61%
Music, Movies and Audio Books	5	\$234,684.64	\$137,426.747	58.56%
Cameras and camcorders	6	\$231,322.53	\$133,446.7616	57.69%
Audio	7	\$80,434.00	\$44,446.9939	55.26%
Games and Toys	8	\$31,514.39	\$16,043.675	50.91%
Total		\$3,550,194.76	\$1,978,968.0633	55.74%

Most sold product by age category

Category	Adult	Old	Young
TV and Video	38.31%	50.07%	11.62%
Music, Movies and Audio Books	34.35%	54.96%	10.69%
Home Appliances	34.18%	58.88%	6.94%
Games and Toys	36.56%	55.34%	8.10%
Computers	36.61%	51.50%	11.89%
Cell phones	36.97%	52.40%	10.63%
Cameras and camcorders	40.76%	52.83%	6.41%
Audio	47.06%	41.12%	11.82%
Total	36.78%	52.95%	10.26%

Bagan Tabel dan Relationships.



1. Skema yang digunakan adalah star schema.
2. Semua dimension table dijamin keunikannya.
3. Semua relationship dijamin keabsahannya (semua key dari table sales tersedia di dimension tabelnya).

Teori fundamental DAX

Berikut adalah konsep DAX yang perlu diketahui untuk memahami rumus dax yang akan digunakan pada dashboard ini. Untuk informasi lebih jelas silahkan kunjungi sqlbi.com.

1. Row Context = Kalkulasi dax dilakukan baris per baris dalam suatu tabel.
2. Filter Context = Filter pada suatu table yang berasal dari visual, slicer ataupun filter pane.
3. Calculate/Calculatable = Satu-satunya fungsi yang dapat mengubah Filter Context.
4. Expanded Table = Suatu table mengambil semua kolom dari tabel lain yang memfilter lewat relationship.
5. Context Transition = Transisi antara row context menjadi filter context. Transisi hanya terjadi jika row context dan fungsi calculate/calculatable terjadi di waktu yang sama.

Penjelasan measure DAX

Visualisasi dengan title “LastYear Sales, Sales and Profit By Month”

Pada visualisasi ini kita akan menggunakan 3 measure yaitu :

Sales Amount LY = `CALCULATE (SUMX (Sales , Sales[Net Price] * Sales[Quantity]), 'Date'[Year] = Max('Date'[Year]) - 1)`

Penjelasan : Disini kita mengubah filter tahun dari tabel date menjadi tahun sebelumnya dari filter tahun yang aktif. Kita menggunakan fungsi calculate untuk mengubahnya, lalu kita menjumlahkan total penjualan dari tabel sales. Tujuan measure ini adalah untuk mendapatkan jumlah penjualan pada tahun sebelumnya. Memfilter tabel date juga memfilter tabel sales, ini terjadi karna relationship antara date ke sales yang membuat sales mengalami ‘expanded table’.

Sales Amount = `SUMX (Sales , Sales[Net Price] * Sales[Quantity])`

Penjelasan : Menjumlahkan total penjualan dari tabel sales sesuai dengan filter context yang aktif.

Profit Amount = `SUMX (Sales , Sales[Profit] * Sales[Quantity])`

Penjelasan : Menjumlahkan total profit dari tabel sales sesuai dengan filter context yang aktif.

Penjelasan measure DAX

Visualisasi dengan title “Sales by Store Country”

Pada visualisasi ini kita akan menggunakan 1 measure yaitu :

Sales Domination =

```
Var A = Calculate ( Sum(Sales[Net Price]) , all ( Store[Country] ) )  
RETURN Sum(Sales[Net Price]) / A
```

Penjelasan : Variable A untuk menampung nilai total penjualan sales menghiraukan semua filter context pada store[country]. Lalu kita membagi total penjualan sales sesuai dengan filter context pada store[country] dengan A. Note: saat menghitung total sales kita tidak menggunakan quantity melainkan langsung memakai net price dikarenakan kita hanya menghitung persentase dan bukan nilai asli.

Penjelasan measure DAX

Visualisasi dengan title “Online Store Customer Order Count”

Pada visualisasi ini kita akan menggunakan 1 measure yaitu :

Online Store Customer Order Count =

```
Calculate ( DISTINCTCOUNT(Sales[Order Number]) , Store[Name] = "Online Store" )
```

Penjelasan : Disini kita menghitung total count dari sales[ordernumber] khusus hanya untuk “Online Store”. Kita menggunakan distinct count dikarenakan dalam tabel sales , order number terduplikat sebanyak jumlah dari item yang terdapat di order number tersebut. Lagi-lagi kita menggunakan calculate untuk mengubah filter context sales table untuk menunjukan hanya online store.

Penjelasan measure DAX

Visualisasi dengan title “Customer buy more than once by country”

Pada visualisasi ini kita akan menggunakan 1 measure yaitu :

```
Customerreturn =  
VAR sum1 = SUMMARIZE(Sales, [Order Number],[CustomerKey])  
VAR totalcust = DISTINCT ( Sales[CustomerKey] )  
VAR totalcustreturn = COUNTROWS (FILTER (totalcust,  
    VAR custkey = [CustomerKey]  
    VAR tabel1 = COUNTROWS (FILTER ( sum1, [CustomerKey] = custkey ))  
    RETURN IF (tabel1 > 1, TRUE, FALSE) ) )  
RETURN totalcustreturn / countrows(totalcust)
```

Goals : Goals dari measure ini adalah untuk menghitung berapa persen customer yang Kembali per jumlah customer. Customer yang melakukan pembelian lebih dari 1 x atau lebih dari 1 PO / Order Number dihitung sebagai customer return.

Penjelasan : VAR sum1 mengandung sebuah tabel berisi kolom sales[order number] dan sales [customer key] dalam bentuk UNIK / NON-DUPLICATE, fungsi summarize mereturn kolom dalam sebuah tabel dan mengubah kolom2 tersebut menjadi tabel dengan nilai kombinasi yang unik. VAR totalcust mengandung tabel yang berisi nilai sales[customerkey] dalam bentuk unik (tabel ini akan digunakan untuk mengecek apakah customer tersebut merupakan customer return atau tidak. VAR totalcustreturn mengandung nilai berapa customer yang return, cara mengeceknya adalah dengan memfilter tabel sum1 sesuai dengan customer yang ada pada totalcust, jika rows dari hasil filter tabel sum1 tersebut lebih dari 1, itu artinya customer tersebut tercatat melakukan pembelian lebih dari 1 x (jumlah order number dari customer tersebut lebih dari 1). Terakhir, kita akan mereturn jumlah customer return per total keseluruhan jumlah customer.

Penjelasan measure DAX

Visualisasi dengan title “TotalStores”

Pada visualisasi ini kita akan menggunakan 1 measure yaitu :

Stores = `DISTINCTCOUNT` (Store[Store Code])

Penjelasan : Menghitung jumlah toko berdasarkan store code.

Penjelasan measure DAX

Visualisasi dengan title “Delivery days by Online Store”

Pada visualisasi ini kita akan menggunakan 1 measure yaitu :

```
DeliveryDur =  
VAR tampung = calculate ( distinctcount(Sales[Order Number]), Sales[Days Duration] <=  
MAX('Day'[Days]), Store[Name] = "Online Store" )  
VAR nilaitotal = calculate ( distinctcount(Sales[Order Number]) , Store[Name] = "Online  
Store")  
VAR kalkulasi = tampung / nilaitotal  
VAR prevalue = calculate ( max('Day'[Days]), 'Day'[Days] < selectedvalue('Day'[Days]))  
VAR pretampung = calculate ( distinctcount(Sales[Order Number]), Sales[Days Duration]  
<= prevalue, Store[Name] = "Online Store" )  
RETURN IF ( PRETAMPUNG / NILAITOTAL = 1 , BLANK() , KALKULASI)
```

Goals : Goals dari measure ini adalah untuk menghitung lama persentase pengiriman barang per customer country.

Penjelasan : Measure ini terdiri dari 2 bagian, untuk lebih mudah mari kita mengeksplor bagian 1 terlebih dahulu. Sambung ke slide berikutnya.

Penjelasan measure DAX

Visualisasi dengan title “Delivery days by Online Store”

Bagian 1 :

```
VAR tampung = calculate ( distinctcount(Sales[Order Number]), Sales[Days Duration] <=
MAX('Day'[Days]), Store[Name] = "Online Store" )
VAR nilaitotal = calculate ( distinctcount(Sales[Order Number]) , Store[Name] = "Online
Store")
VAR kalkulasi = tampung / nilaitotal
RETURN Kalkulasi
```

Penjelasan : VAR tampung mengandung jumlah orderan dimana sales [days duration / lama pengiriman] dibawah atau sama dengan hari pengiriman di filter context. VAR nilaitotal mengandung jumlah orderan total dalam sales tabel. Cukup bagi var tampung dengan nilai total maka akan menghasilkan persentase lama pengiriman barang. Measure diatas dapat langsung dipakai ke dalam visualisasi namun nilai 100% akan ditampilkan secara duplicate di hari pengiriman selanjutnya. Jika nilai 100% tersebut hanya ingin ditampilkan sekali, maka kita harus membuat measure baru yang mengecek nilai hari pengiriman sebelumnya apakah sama dengan 100%, jika sama maka return blank, jadi sebenarnya measure DeliveryDur hanya gabungan dari 2 measure yang sama, hanya filter saja yang berbeda.

Penjelasan measure DAX

Visualisasi dengan title “Most Sold Products”

Pada visualisasi ini kita akan menggunakan 1 measure yaitu :

ProductRank =

```
Var tampung = SELECTEDVALUE ( 'Product'[Category] )
Var table1 = ADDCOLUMNS( ALL( 'Product'[Category] ), "Sales", Calculate( Sumx( Sales ,
Sales[Net Price] * Sales[Quantity] ) ) )
Var table2 = ADDCOLUMNS( table1, "Rank", RANKX( table1, [sales] ) )
Var table3 = FILTER ( table2 , [Category] = tampung )
Return MAXX ( TABLE3 , [Rank] )
```

Penjelasan : VAR tampung berisi value scalar yaitu product[category] yang terlihat dalam filter konteks saat ini. SELECTEDVALUE adalah fungsi untuk mereturn satu-satunya nilai dalam sebuah kolom dalam bentuk scalar. VAR table1 berisi tabel dengan kolom pertama merupakan list dari semua product[category] dan kolom kedua merupakan jumlah penjumlahan dari tabel sales per filter berdasarkan product[category]. Bisa dilihat bahwa pada table1, fungsi addcolumns merupakan sebuah iterator dimana konsep row context hadir dan juga terdapat fungsi calculate dalam expressionnya. Calculate + row context= context transition, itulah mengapa sumx dari table sales akan menghasilkan nilai yang berbeda di tiap rowsnya karena setiap rows filternya juga berbeda-beda. VAR table2 berisi 3 kolom yaitu pada kolom 1 dan 2 merupakan tabel1, dan dikolom terakhir berisi nilai rank atau urutan tiap kategori berdasarkan total salesnya. Fungsi RANK bekerja dengan mengiterate tabel argument pertama lalu mencocokkannya dengan nilai dari row context tabel diluar, lalu menentukan urutan berdasarkan total sales.

```
Profit Amount = SUMX ( Sales , Sales[Profit] * Sales[Quantity] )
```

```
Sales Amount = SUMX ( Sales , Sales[Net Price] * Sales[Quantity] )
```

Penjelasan measure DAX

Visualisasi dengan title “Most Sold Products by Age Category”

Pada visualisasi ini kita akan menggunakan 1 measure yaitu :

`%AgeBuy = [Sales Amount] / CALCULATE ([Sales Amount] , ALL(Customer[Category]))`

Penjelasan : Menghitung jumlah sales berdasarkan filter kategori umur, dibagi semua total penjualan sales. Fungsi All dalam calculate berfungsi untuk melepas semua filter yang ada pada customer[category]. Melepas filter pada tabel customer juga melepas filter pada tabel sales berkat adanya ‘expanded table’